Statistical Procedures Using SAS

Introduction

This handout illustrates commands for some of the more commonly used statistical procedures with SAS.  
Independent samples t-tests

Proc ttest can be used to carry out one-sample, paired sample, or independent samples t-tests. The syntax below is to carry out an independent samples t-test to compare the mean of SALARY for females vs. males:

proc ttest data=sasdata2.employee2;

  class gender;

  var salary;

run;

The output from this procedure is shown below:

                                   The TTEST Procedure

                           Variable:  salary  (Current Salary)

      gender           N        Mean     Std Dev     Std Err     Minimum     Maximum

      f              216     26031.9      7558.0       514.3     15750.0     58125.0

      m              258     41441.8     19499.2      1214.0     19650.0      135000

      Diff (1-2)            -15409.9     15265.9      1407.9

 gender       Method              Mean      95% CL Mean       Std Dev     95% CL Std Dev

 f                             26031.9    25018.3  27045.6     7558.0     6906.2   8346.8

 m                             41441.8    39051.2  43832.4    19499.2    17949.3  21344.3

 Diff (1-2)   Pooled          -15409.9   -18176.4 -12643.3    15265.9    14351.1  16306.1

 Diff (1-2)   Satterthwaite   -15409.9   -18003.0 -12816.7

               Method           Variances        DF    t Value    Pr > |t|

               Pooled           Equal           472     -10.95      <.0001

               Satterthwaite    Unequal      344.26     -11.69      <.0001

                                  Equality of Variances

                    Method      Num DF    Den DF    F Value    Pr > F

                    Folded F       257       215       6.66    <.0001

If you want to check on the distribution of Salary for males and females, you can use Proc Univariate.

proc univariate data=sasdata2.employee2;

  var salary;

  class gender;

  histogram; 
run;

Because it looks like salary is highly skewed, you might want to use a log transformation of salary to compare the two genders. Proc ttest has the dist=lognormal option to accompllish this:

proc ttest data=sasdata2.employee2 dist=lognormal;

  class gender;

  var salary ;

run;

The output from this procedure shows that the geometric mean and coefficient of variation are reported, rather than the arithmetic mean and standard deviation.

                         Variable:  salary  (Current Salary)

                                   Geometric     Coefficient

            gender            N         Mean    of Variation     Minimum     Maximum

            Female          216      25146.1          0.2582     15750.0     58125.0

            Male            258      37972.2          0.4149     19650.0      135000

            Ratio (1/2)               0.6622          0.3505

                               Geometric                        Coefficient

 gender        Method               Mean      95% CL Mean      of Variation       95% CL CV

 Female                          25146.1    24303.8  26017.5         0.2582     0.2353   0.2862

 Male                            37972.2    36161.3  39873.8         0.4149     0.3796   0.4579

 Ratio (1/2)   Pooled             0.6622     0.6226   0.7044         0.3505     0.3284   0.3760

 Ratio (1/2)   Satterthwaite      0.6622     0.6240   0.7028

                                  Coefficients

                 Method           of Variation        DF    t Value    Pr > |t|

                 Pooled           Equal              472     -13.13      <.0001

                 Satterthwaite    Unequal          442.4     -13.63      <.0001

                                     Equality of Variances

                       Method      Num DF    Den DF    F Value    Pr > F

                       Folded F       257       215       2.46    <.0001

To get an independent samples t-test within each job category, use a BY statement, after sorting by jobcat.

proc sort data=sasdata2.employee;

  by jobcat;

run;

proc ttest data=sasdata2.employee;

  by jobcat;

  class gender;

  var salary;

run;

Correlation

Proc corr can be used to calculate correlations for several variables:

proc corr data=sasdata2.employee;

  var salary salbegin educ;

run;

                                The CORR Procedure

                        3  Variables:    salary   salbegin educ

                                    Simple Statistics

Variable           N          Mean       Std Dev           Sum       Minimum       Maximum

salary           474         34420         17076      16314875         15750        135000

salbegin         474         17016          7871       8065625          9000         79980

educ             474      13.49156       2.88485          6395       8.00000      21.00000

                                    Simple Statistics

                          Variable    Label

                          salary      Current Salary

                          salbegin    Beginning Salary

                          educ        Educational Level (years)

                             Pearson Correlation Coefficients, N = 474
                               Prob > |r| under H0: Rho=0
                                            salary      salbegin          educ

           salary                          1.00000       0.88012       0.66056

           Current Salary                                 <.0001        <.0001

           salbegin                        0.88012       1.00000       0.63320

           Beginning Salary                 <.0001                      <.0001

           educ                            0.66056       0.63320       1.00000

           Educational Level (years)        <.0001        <.0001

Linear regression

You can fit a linear regression model using Proc Reg:

proc reg data=sasdata2.employee2;

  model salary = salbegin educ jobdum2 jobdum3 prevexp female;

  output out=regdat p=predict r=resid rstudent=rstudent;

run; quit;

Note that the output dataset that we created, REGDAT, has all the original observations and variables in it, plus the new variables Predict, Resid, and Rstudent. 

Output from the linear regression model is shown below:

                                  The REG Procedure

                                      Model: MODEL1

                        Dependent Variable: salary Current Salary

                         Number of Observations Read         474

                         Number of Observations Used         474

                                   Analysis of Variance

                                          Sum of           Mean

      Source                   DF        Squares         Square    F Value    Pr > F

      Model                     6     1.15239E11    19206503793     395.52    <.0001

      Error                   467    22677472676       48559899

      Corrected Total         473    1.379165E11

                   Root MSE           6968.49328    R-Square     0.8356

                   Dependent Mean          34420    Adj R-Sq     0.8335

                   Coeff Var            20.24573

                                   Parameter Estimates
                                                Parameter     Standard

 Variable   Label                         DF     Estimate        Error  t Value  Pr > |t|

 Intercept  Intercept                      1   5333.10875   2337.45787     2.28    0.0230

 salbegin   Beginning Salary               1      1.31359      0.07433    17.67    <.0001

 educ       Educational Level (years)      1    548.90277    163.27562     3.36    0.0008

 jobdum2                                   1   6764.00748   1666.58592     4.06    <.0001

 jobdum3                                   1        11389   1394.92854     8.16    <.0001

 prevexp    Previous Experience (months)   1    -21.98825      3.64720    -6.03    <.0001

 female                                    1  -2122.17197    775.86768    -2.74    0.0065
Check for normality of the residuals by using Proc Univariate.

proc univariate data=regdat;

  var rstudent;

  histogram / normal;
  qqplot / normal(mu=est sigma=est);
run;

We note that the distribution of the residuals is highly skewed. We may want to transform salary using the natural log. The commands below show how Logsalary can be created to be used in the regression. Note that to create a new variable, we need to use a data step. Submit these commands and check the residuals from this new regression model.
data temp;

  set sasdata2.employee2;

  logsalary = log(salary);

run;

proc reg data=temp;

  model logsalary = salbegin educ jobdum2 jobdum3 prevexp female;

  output out=regdat2 p=predict r=resid rstudent=rstudent;

run; quit;

proc univariate data=regdat2;

  var rstudent;

  histogram / normal;

  qqplot / normal(mu=est sigma=est);

run;
Cross-tabulations

You can carry out a Pearson Chi-square test of independence using Proc Freq. This procedure is extremely versatile and flexible, and has many options available.

proc freq data=sasdata2.employee2;

   tables gender*jobcat / chisq;

run;

                                                     The FREQ Procedure
                                Table of gender by jobcat

                       gender(Gender)

                                 jobcat(Employment Category)

                       Frequency|

                       Percent  |

                       Row Pct  |

                       Col Pct  |       1|       2|       3|  Total

                       ---------+--------+--------+--------+

                       f        |    206 |      0 |     10 |    216

                                |  43.46 |   0.00 |   2.11 |  45.57

                                |  95.37 |   0.00 |   4.63 |

                                |  56.75 |   0.00 |  11.90 |

                       ---------+--------+--------+--------+

                       m        |    157 |     27 |     74 |    258

                                |  33.12 |   5.70 |  15.61 |  54.43

                                |  60.85 |  10.47 |  28.68 |

                                |  43.25 | 100.00 |  88.10 |

                       ---------+--------+--------+--------+

                       Total         363       27       84      474

                                   76.58     5.70    17.72   100.00

                         Statistics for Table of gender by jobcat

                  Statistic                     DF       Value      Prob

                  ------------------------------------------------------

                  Chi-Square                     2     79.2771    <.0001

                  Likelihood Ratio Chi-Square    2     95.4629    <.0001

                  Mantel-Haenszel Chi-Square     1     67.4626    <.0001

                  Phi Coefficient                       0.4090

                  Contingency Coefficient               0.3785

                  Cramer's V                            0.4090

                                    Sample Size = 474

You can get an exact test for this by using an Exact statement. In this case, we requested Fisher’s exact test, but exact p-values for other statistics can be requested:
proc freq data=sasdata2.employee;

   tables gender*jobcat / chisq;

   exact fisher;

run;
In the output below, be sure to read the last p-value at the bottom of the output for Fisher’s exact test.

                                      Fisher's Exact Test

                               ----------------------------------

                               Table Probability (P)    2.854E-22

                               Pr <= P                  5.756E-21

                                        Sample Size = 474

If  your problem is large, you may wish to get a Monte Carlo simulation for the p-value, based on 10, 000 tables. To do this use the following syntax. Seed=0 will use a random seed for the process based on the clock time when you run the procedure.

  proc freq data=sasdata2.employee;

   tables gender*jobcat / chisq;

   exact fisher / mc seed=0;

  run;
Partial output from this procedure is shown below:

                                       The FREQ Procedure

                            Statistics for Table of gender by jobcat

                                      Fisher's Exact Test

                               ----------------------------------

                               Table Probability (P)    2.854E-22

                             Monte Carlo Estimate for the Exact Test

                               Pr <= P                     0.0000

                               99% Lower Conf Limit        0.0000

                               99% Upper Conf Limit     4.604E-04

                               Number of Samples            10000

                               Initial Seed             445615001

                                        Sample Size = 474

Each time the procedure is run using this syntax, you will get different answers. If you wish to get the same result, simply use the Initial Seed value reported by SAS in the output in your Exact statement.

proc freq data=sasdata2.employee;

   tables gender*jobcat / chisq;

   exact fisher / mc seed=445615001;

run;

